Занятие I

Тема. Тeкстовые файлы, их описание и основные отличия от типизированных файлов. 

Наряду с описанными типами файлов Pascal имеет средства взаимодействия с файлами несколько иной структуры–так называемыми текстовыми файлами. Введение текстовых файлов несколько нарушает стройность языка, однако позволяет использовать Pascal при программировании широкого класса задач, имеющих нечисловой характер и связанных с обработкой текстовой информации.

Во многих версиях языка допускается хранение файлов на диске как символьных данных. При считывании файла в оперативную память машины символы файла преобразуются в тот тип данных, который объявлен в программе. Файлы символьных данных называются текстовыми файлами. Текстовые файлы имеют тип text.

Структура текстовых файлов отличается от структуры обычных файлов (которые представляют из себя линейную последовательность элементов одного типа) тем, что содержимое текстового файла рассматривается как последовательность строк переменной длины, разделённых специальной комбинацией, называемой "конец строки". Как правило, эта комбинация строится из управляющего кода "перевода каретки" (CR, Carriage Return, символ #13), за которым, возможно, следует управляющий код "перевод строки" (LF, Line Feed, символ #10). Признаком конца строки считается нажатие клавиши ввода. 

Текстовый файл завершается специальным кодом "конец файла" (символ #26). В большинстве случаев знание конкретной кодировки управляющих символов не обязательно ввиду наличия файловых операций, автоматически учитывающих эти символы.

Таким образом, текстовый файл структурно несколько похож на "файл из байтов" (file of byte) с той разницей, что в нем, помимо содержательной информации, встречаются символы специального назначения. 

Его можно схематически представить в следующем виде:

Описанная структура текстовых файлов хорошо согласуется с интуитивно понимаемым построением текстовой информации и полностью совпадает со стандартной структурой текстов, принятой в MS-DOS, используемой во многих текстовых редакторах, понимаемой компиляторами с языков программирования и т.д.

С каждым файлом на диске должна быть связана файловая переменная, которая описывается в соответствии с типом файла на диске. Представителем же текстового файла в Pascal-программе является переменная файлового типа, которая должна быть описана с указанием стандартного типа text:

Var

   TextFile : text;

Примечание. Слово text не является зарезервированным словом, а считается идентификатором стандартного типа, наряду с идентификаторами integer, real и т.д.

Обращение к файлу в дальнейшем идёт через файловую переменную.

Далее доступ к файлу требуется открыть. Открыть любой файл можно на чтение и на запись. Для этого существуют процедуры reset, rewrite.

К примеру, пусть на диске создан текстовой файл text.txt.

Для Turbo Pascal описание и связывание файловой переменной  f с файлом text.txt, будет выглядеть так :

Var

  f: text;

Begin

   assign(f, 'd:\tp7\bin\text.txt'); {Полный путь до файла }

   reset(f);{Открыть на чтение]

 . . .

End.

assign([файл. пер. ], [полный путь до файла на диске])–связывает файл на диске с файловой переменной f типа текст.

Примечание. Процедура assign не должна использоваться для открытого файла.

Когда имя файла на диске не указывается, то параметр f оказывается связанным со стандартным вводом/выводом. Если присвоено пустое имя, то после обращения к reset(f) f будет указывать на стандартный файл ввода, а после обращения к rewrite(f) f будет указывать на стандартный файл вывода.


reset([ файловая переменная ]); –открывает файл на чтение.


Ввод-вывод для текстовых файлов подчиняется тем же общим правилам, что и для обычных типизированных файлов; однако имеется несколько важных особенностей.

Во-первых, для одного текстового файла нельзя одновременно производить операции и ввода, и вывода. Это означает, что после открытия текстового файла процедурой reset возможно только чтение информации из файла, после процедуры rewrite – только запись в файл.

Во-вторых, обмены с текстовыми файлами всегда являются строго последовательными, то есть после чтения из файла элемента с порядковым номером N следующая операция чтения даст элемент с номером N+1. Иными словами, прямой доступ к любому элементу текстового файла невозможен; для текстовых файлов не допускаются вызовы Seek, FilePos, FileSize. 

Под чтением файла понимают ввод данных из внешнего файла, находящегося на диске, в оперативную память машины. Данные файла становятся доступными программе. Внешний файл, из которого читаются данные, часто называют входным файлом.

Базовой техникой обменов с текстовыми файлами является посимвольный ввод-вывод. При этом производится чтение или запись всех символов, как информационных, так и специальных.

Покажем простую программу, выполняющую чтение некоторого текста. Эта программа выводит на экран последовательность кодов символов, составляющих файл text.txt.

Program TextFile1;

Var

    f : text;
 {Читаемый текст}

    S : char; {Очередной прочитанный символ}

Begin

    assign(f, 'text.txt'); {Связываем файл text.txt с переменной f}

    reset(f,); {Открываем файл text.txt для чтения}

    while not Eof(f) do{Пока нет конца файла делай...}

        begin

           readln(f, S);{Читаем из файла очередной символ в переменную а}

           writeln(a);{Печатаем код символа на экран}

        end;

    close(f);{Закрываем файл}

    readln
End.

Задание. Наберите текст программы и запустите программу на выполнение. Просмотрите результат работы программы. Найдите выведенные на экран коды специальных символов. 

Еще одной особенностью работы с текстовыми файлами является возможность непосредственных обменов значениями различных базовых типов (для обычных типизированных файлов тип параметров read и write должен совпадать с базовым типом файла). Так, в текстовый файл можно записать6 например, целое или вещественное число. При этом его внутреннее представление будет автоматически преобразовано в строчку символов, образующих изображение этого числа.

Рассмотрите простую программу, выполняющую чтение из текстового файла целых чисел и вывод на печать только четных чисел. 

Program TextFile2;

Var

    f : text;

    Put : string;

    a : integer;

Begin

    Put := 'D:\TP7\BIN\Primer2'; {Полный путь до файла }

    assign(f, Put); {Связываем файл с переменной f}

    reset(f); {Открываем файл на чтение.}

    while not Eof(f) do{Пока нет конца файла делай...}

        begin

            readln(f, a);{Считываем число в переменную а}

            if not odd(a) {Если число нечетное,}

               then
                   writeln(a);{То выводим его на экран}

      end;

    close(f);

    readln
End.

Задание. Напишите программу, с помощью которой можно сформировать текстовый файл, содержащий 

- значения целого типа;

- значения вещественного типа;

- значения булева типа.

Для описанных выше случаев записи произвольной информации в текстовый файл в языке имеются дополнительные возможности, которые заключаются в задании размера поля записи. Если после записываемой переменной или выражения поместить символ двоеточия, а после него – любое выражение целого типа, то для выводимого значения будет отведено поле, размер которого (число символов) будет равен значению выражения. Следующая тривиальная программа наглядно показывает действие указателя поля:

Program TextFile3;

Var

   f : text;

   V : real;

   i : word;

   Put : string;

Begin

    Put := 'D:\TP7\BIN\Primer3'; {Полный путь до файла }

    assign(f, Put); {Связываем файл с переменной f}

    rewrite(f); {Открываем файл на запись}

    V := 123.456;
    write(f, V,#13#10);{выводим без указателя поля на экран}

    for i := 8 to 12 do

       write(f, V:i,#13#10);{динамически изменяем размер поля}

    close(f);

End.

Задание. Наберите программу, дополните ее выводом содержимого текстового файла на экран. Проанализируйте полученные результаты выполнения программы.

Управление размещением значений в текстовых файлах очень удобно при формировании структурированных файлов (списков, таблиц и т.п.), так как если изображение значения меньше размера поля, то оно всегда "прижимается" к его правому краю. 

Кроме того, для вещественных типов имеется дополнительная возможность, позволяющая выводить число в формате с фиксированной точкой, что более наглядно по сравнению с форматом с плавающей точкой, который предусмотрен по умолчанию. Если после указателя размера поля задать через двоеточие еще одно выражение целого типа, то оно будет интерпретироваться как указание числа позиций для дробной части числа.

Занятие II

Тема. Способы обмена с текстовыми файлами.

На этом занятии мы обобщим уже известные операции чтения и записи текстовых файлов и остановимся на специфичных операциях обмена информацией между программой и текстовым файлом.

I Операции чтения из файла

reset(f) - открывает существующий файл на чтение. Файловая переменная должна быть связана с внешним файлом с помощью процедуры assign. 

Если существующий файл уже открыт, то он закрывается, а затем открывается вновь. Текущая позиция в файле устанавливается на начало файла. 

Если f было присвоено пустое имя (например, assign((f),' '), то после обращения к процедуре Rezet(f) будет ссылкой на стандартный файл ввода (канал 0). 

Текстовой файл становится доступным только по чтению. 

При указании директивы компилятора {$I–} функция IoResult будет возвращать значение 0 в том случае, если операция завершилась успешно, и ненулевой код ошибки в противном случае.

readln([f : text],[v1,v2,....vn]) – выполняет процедуру read, затем переходит к следующей строке файла. 

Процедура readln является расширением процедуры read и определена для текстовых файлов. Вызов readln(f) без параметров приводит к перемещению текущей позиции файла на начало следующей строки, если она имеется, в противном случае происходит переход к концу файла. Процедура readln без параметров полностью соответствует стандартному вводу. 

При указании директивы компилятора {$I-} функция IoResult будет возвращать значение 0 в том случае, если операция завершилась успешно, и ненулевой код ошибки в противном случае. 

Примечание: Функция работает только для тектовых файлов, включая стандартный ввод. Файл должен быть открыт для ввода.

Процедура readln является очень удобным аналогом read. В случае использования readln после чтения из файла очередной порции символов (и, быть может, преобразования их в значение подходящего типа) текущий указатель файла будет перемещен на начало его следующей строки. Иными словами, часть строки после прочитанного значения будет пропущена.

Очень часто используется следующий способ чтения из текстового файла. В качестве параметра процедуры readln задается переменная типа string; в этом случае вся очередная строка файла целиком считывается в поданную переменную, длина которой автоматически устанавливается равной длине считанной строки. Полученная из файла строка далее может быть обработана так, как это необходимо. Важно отметить, что при таком способе производится чтение только "значащих" символов строки; завершающиеся символы в данном случае играют роль межстрочных разделителей и не считываются в строковую переменную. После чтения строки текущий указатель файла устанавливается на начало следующей строки.

II Операции записи в файл

rewrite(f) - cоздаёт и открывает новый файл. Файловая переменная должна быть связана с внешним файлом с помощью процедуры assign. 

Если внешний файл уже существует, то он удаляется и на его месте создаётся новый пустой файл. Если файл уже открыт, то он закрывается, а затем открывается вновь.Текущая позиция в файле устанавливается на начало файла. 

Если f было присвоено пустое имя (например, assign((f),' '), то после обращения к процедуре rewrite(f) будет ссылкой на стандартный файл вывода (канал 1). 

Текстовой файл становится доступным только по записи. 

write([f : text],[v1,v2,....vn]) – записывает одно или более значений из одной или более переменных в текстовой файл.

Если параметр f не указан, то подразумевается использование стандартной файловой переменной Output. Каждый параметр v является вводимым выражением, значение которого должно быть записано в файл. Каждое вводимое выражение должно быть символьного, целого, вещественного, строкового или булевого типа.

Параметр v имеет вид:


<вводимое выражение : мин.размер : десятичных знаков>

Например: 

Var 

    a:real;

 . . . . . . . 

write(f, a: 5: 2)

Такая запись означает, что мы в файл записываем действительное (не целое) число а, размером 5 знаков, 2 знака под дробную часть.
writeln([f : text],[v1,v2,....vn]) – выполняет процедуру write, а затем записывает в файл метку конца строки (перевод строки). 

При вызове данной процедуры без параметров writeln(f), в файл записывается метка конца строки. Процедура writeln без параметров полностью соответствует стандартному выводу на экран.

Примечание: файл должен быть открыт для вывода.

III Логическая функция Eoln

Часто для обработки текстовых файлов используется специфичная для них функция Eoln, позволяющая определить достигнут ли конец строки. Если достигнут – значение функции равно True, а если нет – False. Таким образом, для анализа конкретных символов строк файла можно применить вложенный цикл типа:

while not Eof(NameFale) do {пока нет конца файла NameFale делай}

   while not Eoln(NameFale) do {пока нет конца строки файла NameFale делай}

       begin

           {группа операторов обработки символов очередной строки}

       end;

IV Процедура открытия файла для дополнения

append(f : Тext) – процедура открывает существующий файл для присоединения. Если файл уже открыт, то он сначала закрывается, а затем открывается заново. Текущая позиция устанавливается на конец файла.

Если в последнем блоке файла размером 128 байтов присутствует символ Сtrl+Z (26 в коде ASCII), то текущая позиция устанавливается в файле таким образом, что при записи первым в блоке будет "затираться" символ Сtrl+Z.

Если переменной f было присвоено пустое имя assign(f,'  '), то после обращения к процедуре append f будет указывать на стандартный выходной файл.

После обращения к append файл f становится доступным только по записи и Eof(f) принимает всегда значение True(истина). 

Рассмотрите несколько примеров простых программ.

Program TextFile4;

{Задача. Дан файл Primer1 на диске. Считать из него информацию и записать во вновь созданный файл Primer2 }

Var

    f, t: text;

    Put1, Put2, d: string;

    s : char;

Begin

    Put1 := 'D:\TP7\BIN\Primer1';{Путь до файла Primer на диске}

    assign(f, Put1 );

    reset(f);{Открыть файл c файловой пер. f на чтение }

    Put2 := 'D:\TP7\BIN\Primer2';{Путь где будет создан файл}

    assign(t, Put2);

    rewrite(t);{ Открыть файл c файловой пер. t на создание и запись }

    while Not Eof(f) Do

        begin

            while Not Eoln(f) do

                begin

                    read(f, s); {Cчитывать из f в переменную s}

                    write(t, s);{ Записывать в t значение s}

                end;

             readln(f); {Переходим к чтению следующей строки файла f}

             writeln(t);{Переходим к записи следующей строки файла t}

        end;

    close(t);{Закрытие файла Primer1}

    close(f);{Закрытие файла Primer2}

End.

Program TextFile5;

{Задача. Дан файл Primer2 со скобками. Проверить численное соответствие закрытых и открытых скобок}

Var

    f : Тext;

    Put, s : string;

    a, i : integer;

Begin

    a := 0; {Обнуляем счетчик скобок}

    Put := 'D:\TP7\BIN\Primer2'; {Полный путь до файла }

    assign(t, Put);

    reset(f); {Открываем файл на чтение.}

    while Not Eof(f) do {Пока нет конца файла делай...}

        begin

           readln(f, s); {Считываем cтроку в переменную s}

           for i := 1 to Length(s) do {Просматриваем строку до конца}

              begin

                 if s[i] = '(' {Если встретилась открытая скобка,}

                   then
                      Inc(a,1);{то счетчик увеличиваем на 1}

                 if s[i] = ')' {Если встретилась закрытая скобка,}

                   then
                      Dec(a,1);{то счетчик уменьшаем на 1}

              end;

        end;

    close(f);{Закрытие файла}

    if a <> 0{Если скобок одинаковое кол-во, то а=0}

       then
          writeln('No')

       else
          writeln('Yes');

    readln
End.

Program TextFile6;

{Задача. Проиллюстрировать работу процедуры append.}

Var

    f: text;

Begin

    assign(f, 'text.txt'); {создаем текстовый файл}

    rewrite(f);{ Открыть файл c файловой пер. t на создание и запись }

    writeln(f, 'Исходный текст');{ Записывать в него cтроку}

    close(f);{Закрываем сформированный файл}

    append(f); {Вновь открываем этот же файл для добавления}

    writeln(f, 'Добавляемый текст');{ Пишем в конец файла cтроку}

    close(f);{Закрываем файл}

End.

Задачи для самостоятельной работы

1. Создать текстовый файл, содержащий произвольное количество гласных букв русского алфавита. Длину каждой строки должен определять пользователь. Просчитать количество символов каждой буквы и дописать в файл полученную информацию.

2. Создать текстовый файл, содержащий произвольное количество букв русского алфавита. Длину каждой строки должен определять пользователь. Просчитать количество символов гласных и согласных букв и дописать в файл полученную информацию.

3. Создать текстовый файл, содержащий произвольное количество символов. Длину каждой строки должен определять пользователь. Просчитать количество символов каждой буквы русского алфавита и дописать в файл полученную информацию.

4. Создать текстовый файл, содержащий произвольное количество символов. Длину каждой строки должен определять пользователь. Просчитать количество цифр и дописать в файл полученную информацию.

5. Создать текстовый файл, содержащий произвольное количество символов. Длину каждой строки должен определять пользователь. Просчитать количество символов каждой буквы латинского алфавита и дописать в файл полученную информацию.

Занятие III
Тема. Стандартные текстовые файлы Input и Output. Примеры задач

В Паскале существуют два стандартных текстовых файла Input и Output. Эти файлы считаются известными в любой Pascal-программе (иными словами, они описаны в стандартном модуле System). Они обозначают (по терминологии MS-DOS) соответственно стандартный файл ввода и стандартный файл вывода. Обычно эти стандартные файлы связаны с конкретными физическими устройствами компьютера. Так, файловая переменная Input связана с клавиатурой, файловая переменная Output – с экраном дисплея. эти файлы считаются заранее открытыми, а соответствующие идентификаторы можно использовать в операциях ввода-вывода.

Рассмотрим, например, следующий оператор:

writeln (Output, 'Результат равен ', (X+Y)*2)

В соответствии с общими правилами, этот оператор выведет значения двух последних операндов в текущую строку заданного файла, а затем произведет переход к следующей строке. В применении к стандартному файлу Output эти действия будут выглядеть как появление в текущей строчке дисплея литеральных изображений указанных значений, после чего курсор будет перемещен в первую позицию следующей строки.

Аналогично, оператор read (Input, X1, X2); будет выполняться таким образом: система перейдет в состояние ожидания ввода с клавиатуры двух значений. Типы вводимых значений должны совпадать с типами переменных Х1 и Х2. Эти значения при вводе должны отделяться друг от друга одним или несколькими пробелами, а ввод должен быть завершен нажатием клавиши Enter. В процессе ввода значений набираемые на клавиатуре символы отображаются на экране. После нажатия Enter введенные значения будут присвоены переменным Х1 и Х2, и выполнение программы будет продолжено.

Для стандартных файлов Input и Output допускается сокращенная форма записи операций ввода-вывода. Так, если в процедурах read и readln первый параметр опущен, то по умолчанию подразумевается файл Input. Аналогично, отсутствие в процедурах write и writeln первого параметра означает вывод в стандартный файл Output. Вывод в стандартный файл Output используется очень часто – всегда, когда необходимо выдать некоторую информацию из программы на экран.

В соответствии с общими правилами MS-DOS стандартные файлы ввода-вывода могут быть "переназначены", то есть связаны с другими физическими устройствами или дисковыми файлами. Простейшим способом переназначения является использование для этой цели процедуры assign, например,

assign (Output,'MyFile.out') 

После выполнения такого оператора стандартный файл вывода будет переназначен, то есть файловая переменная Output будет связана с дисковым файлом MyFile.out из текущего каталога. Все операции вывода, явно или неявно работающие с файлом Output, будут выводить информацию в указанный дисковый файл.

Задание. Рассмотрите предложенные программы, наберите их на компьютере, выделите необходимые части алгоритма в подпрограммы, сформулируйте решаемые ими задачи, дополните необходимыми операторами и комментарием. 

Примечание. Если Вы можете решить выше поставленные задачи устно, то поставьте в известность учителя и приготовьтесь к ответу.

Program StringCount;

Uses

   Crt;

Var

   i : integer;

   s : string;

   f : text;

Begin

   ClrScr;

   readln(s);

   assign(f,s);

   reset(f);

   while not Eof(f) do

      begin

         readln(f,s);

         i:=i+1;

      end;

   close(f);

   readln;

End.

Program Zamen;

Uses

   Crt;

Var

   s : string;

   f, d : text;

Begin

   ClrScr;

   readln(s);

   assign(f,s);

   readln(s);

   assign(d,s);

   reset(f);

   rewrite(d);

   while not Eof(f) do

      begin

         readln(f,s);

         while Pos('o',s)>0 do

             s[Pos('o',s)]:='a';

         writeln(d,s);

      end;

   close(f);

   close(d);

   readln;

End.

Program MaxInFile;

Uses

   Crt;

Var

   i, j, r, Code : integer;

   s : string;

   f : text;

Begin

   ClrScr;

   readln(s);

   assign(f,s);

   reset(f);

   while not Eof(f) do

      begin

          readln(f,s);

          j:=0;

          for i:=1 to Length(s) do

              begin

                  Val(s[i],r,Code);

                  if r>j 

                     then 

                        j:=r;

               end;

            writeln(j);

       end;

   close(f);

   readln;

End.

Занятие IV
Тема. Самостоятельное решение задач

Выберите с учителем задачи для самостоятельного решения из предложенного ниже списка.

1. Посчитать количество строк, находящихся между строками максимальной и минимальной длины текстового файла. Вывести эти строки на экран в обратном порядке. Если все строки файла одинаковы, вывести на экран соответствующее сообщение. В программе используйте подпрограммы.

2. Упорядочить лексикографически строки текстового файла. Вывести на экран первоначальный файл и преобразованный. Если все строки файла одинаковы, вывести на экран соответствующее сообщение. В программе используйте подпрограммы.

3. Удалите из текстового файла одинаковые строки. Если в файле нет одинаковых строк, вывести на экран соответствующее сообщение. Вывести на экран первоначальный файл и преобразованный. В программе используйте подпрограммы.

4. Сформировать текстовый файл, содержащий строки исходного файла, встречающиеся в нем один раз. Вывести содержимое файлов на экран. В программе используйте подпрограммы.

5. Найти и вывести на экран самое большое симметричное слово текста. Если симметричных слов в тексте нет, вывести соответствующее сообщение. В программе используйте подпрограммы.

6. В каждом слове текстового файла поменять местами первый и последний символы. Вывести на экран первоначальный файл и преобразованный. В программе используйте подпрограммы.

7. Сравните соответствующие строки двух текстовых файлов. Если строки совпали, то запишите эту строку в результирующий текстовый файл. Вывести на экран данные файлы и полученный. Если полученный файл пуст, вывести соответствующее сообщение. В программе используйте подпрограммы.

8. Создайте текстовый файл, содержащий гласные буквы каждой нечетной строки исходного файла и цифры каждой четной строки файла. В программе используйте подпрограммы.

9. Создайте текстовый файл, содержащий в начале каждой строки гласные буквы соответствующей строки данного текстового файла, а в конце строки – согласные буквы. В середине строки расположите все остальные встретившиеся символы, заключив их в круглые скобки.

11. Удалить из середины каждой строки текстового файла заданное пользователем количество символов. Создайте текстовый файл, содержащий удаленные символы. Вывести на экран первоначальный файл и преобразованный. В программе используйте подпрограммы.

12. Для каждого символа текста указать, сколько раз он встречается в тексте и вывести в текстовый файл слово, в котором этот символ встречается наиболее часто. Если таких слов несколько, вывести 

- первое встретившееся,

- последнее встретившееся,

- все слова. 

В программе используйте подпрограммы.

13. Проанализировав длины строк текстового файла, объедините соседние строки так, чтобы длина строк файла была в заданном пользователем промежутке [а, b]. Если строк для объединения нет, предусмотрите в программе вывод соответствующего сообщения. Вывести на экран первоначальный файл и преобразованный. В программе используйте подпрограммы.

14. Проанализировав созданный текстовый файл, найдите в нем ошибки на правописание чк-чн, нч-нщ. Создайте текстовый файл слов с ошибками. Исправьте ошибки в исходном файле. Если ошибок в файле нет, предусмотрите в программе вывод соответствующего сообщения. Вывести на экран первоначальный файл и преобразованный. При выводе на экран содержания измененного файла, слова, где была ошибка, высветите красным цветом. В программе используйте подпрограммы.

15. Вывести на экран и в текстовый файл строки данного текстового файла, содержащие одинаковые слова. Если таких строк нет, предусмотрите в программе вывод соответствующего сообщения. В программе используйте подпрограммы.

Творческое задание. Придумайте собственную задачу по данной теме и решите ее.

Приготовьте листинги решенных и проверенных учителем задач. Проверьте свои знания, ответив на ниже следующие вопросы.

Контрольные вопросы

1. Что общего в организации работы с текстовыми файлами и типизированными?

2. Какие специфичные для текстовых файлов функции и процедуры Вы знаете?

3. Какую информацию можно хранить в текстовом файле? Сколько способов просмотра текстового файла Вы использовали?

4. Можно ли, пользуясь средствами Norton Commander создать текстовый файл, который можно было бы обработать в Паскаль-программе? Если – да, то какое расширение он должен иметь?

5. Какие процедуры и функции модуля CRT Вы применяли в своих программах? С какой целью?
Для любознательных

Текстовый режим. Модуль CRT. Работа с цветом. Работа с экраном и курсором

Средства управления экраном образуют один из важнейших элементов программирования для IBM-совместимых персональных компьютеров, что обусловлено прежде всего необходимостью организации удобного взаимодействия программы с ее пользователем. Любая сколько нибудь серьезная программа в настоящее время должна иметь достаточно развитый и продуманный экранный "видеообраз", с помощью которого реализуется интерфейс с пользователем.

Рассмотрим общие принципы организации отображения информации на экран дисплея. Формирование изображений на экране производится дисплейным адаптером, который представляет собой специализированный микропроцессор с собственной памятью, регистрами и наборами команд. Интерфейс с адаптером реализуется ядром операционной системы (BIOS), который управляет адаптером, записывая значения в его память, устанавливая состояние регистров и инициируя выполнение команд. Со своей стороны BIOS предоставляет прикладным программам (и операционной системе) возможности управления дисплейным адаптером посредством функций прерывания 10h (видеосервис). Некоторые из них мы рассмотрим.

Заметим, что память дисплейного адаптера включена в общее пространство компьютера и тем самым доступна напрямую и для прикладных программ. Данное обстоятельство весьма существенно, так как содержимое этой памяти непосредственно отображается на экране. Зная правила "кодирования" изображения в памяти адаптера и адреса этой памяти, можно формировать картину на экране, записывая в оперативную память те или иные значения.

Дисплейная аппаратура ориентирована на два режима работы: текстовый и графический. В текстовом режиме рабочее пространство экрана рассматривается как совокупность строк, каждая из которых содержит определенное  число позиций (знакомест). В каждой позиции может быть отображен один из 256 стандартных ASCII-символов. Текстовый режим может иметь несколько вариантов, различающихся количеством и длиной экранных строк (наиболее распространенным является размер 25 строк по 80 символов в каждой). 

Текстовый режим является как сравнительно простым для понимания, так и несложным и эффективным в реализации. Вместе с тем он сильно ограничен по свои возможностям. Необходимая программная поддержка этого режима со стороны операционной системы является минимальной и в то же время позволяет разрабатывать развитые и эффективные интерфейсы.

Модуль CRT

Модуль CRT представляет собой библиотеку функций и процедур, предназначенных для увеличения возможностей текстового ввода-вывода данных. В отличие от стандартного ввода-вывода, когда он осуществляется через операционную систему, подпрограммы модуля CRT работают с BIOS, и даже непосредственно с видеопамятью.

Для того чтобы стали доступны находящиеся в модуле процедуры и функции, необходимо явно указать модуль CRT командой

Uses

   Crt;

Модуль CRT сравнительно невелик и содержит средства, позволяющие:

- устанавливать определенный текстовый режим;

- управлять перемещением курсора на экране;

- выводить на экран информацию;

- задавать цветовые атрибуты выводимых символов;

- организовывать простой оконный дизайн экрана.

Кроме этого, в модуле имеются средства взаимодействия с клавиатурой, процедуры элементарного управления звуком, а также некоторые другие примитивы.

Примечание. Относительная компактность модуля CRT находится в очевидном противоречии с весьма широким спектром возможностей современных адаптеров. Иными словами, за рамками этого модуля остается много средств управления экраном. Поэтому, практически любая программа с развитой экранной компонентой использует, наряду со штатными средствами модуля CRT, дополнительные возможности адаптера, не поддерживаемые этим модулем. Многие сервисные пакеты независимых фирм, ориентированные на Turbo Pascal (например, Power Tools Plus, Object Professional), содержат расширенные варианты модуля CRT. На следующем занятии мы рассмотрим некоторые дополнительные возможности управления экраном, существенные в практической работе.

Итак, мы уже знаем, что при работе с экраном через модуль CRT весь экран разбивается на отдельные строки, а каждая строка ​ на отдельные позиции, в каждую из которых можно поместить один символ. Таким образом экран разбивается на отдельные прямоугольные элементы. 

Рассмотрим допустимые текстовые режимы адаптера и технику их  установки в Turbo Pascal- программах.

В общем случае модуль CRT поддерживает пять базовых вариантов текстового режима и еще несколько дополнительных.    

Для установки одного из текстовых режимов предназначена процедура

Procedure TextMode (Mode : word);

где Mode – код текстового режима, который может принимать следующие значения:

- BW40=0 – 40х25 черно-белый режим для цветного адаптера,

- CO40=1 – 40х25 цветной режим для цветного адаптера,

- BW80=2 – 80х25 черно-белый режим для цветного адаптера,

- CO80=3 – 80х25 цветной режим для цветного адаптера,

- MONO=7 – 80х25 монохромный для черно-белого адаптера,

Например,

TextMode (СО80);

TextMode (7);

Перечисленные режимы действительны для двух наиболее распространенных в настоящее время адаптеров VGA и EGA.

Кроме данных режимов, для VGA/EGA-адаптеров допускаются их модификации, дающие возможность увеличит количество строк, выводимых на экран. Дело в том, что разрешающая способность этих адаптеров (то есть число пикселов по горизонтали и вертикали) позволяет строить символы в виде матриц пикселов 8х14 (для EGA) и 8х16 (для VGA). Иными словами, каждый символ строится из 14 или 16 рядов пикселов. Если же использовать для символов сокращенные матрицы размером 8х8 пикселов, то на экране можно разместить соответственно 43 или 50 строк символов, что дает возможность выводить на экран значительно больше информации.

Чтобы переключиться на уплотненный режим, необходимо подать в процедуру TextMode модифицированный параметр – значение требуемого режима, увеличенное на 256. Для удобства и большей наглядности в модуле имеется соответствующая константа:

- FONT8X8=256 – используется для загружаемого шрифта в режиме 80х43 или 80х50 с адаптерами VGA и EGA.

Рассмотрите пример переключения дисплея в режим цветного отображения с 43/50 строками:

TextMode (СО80+FONT8X8)

Примечание. Многие программные системы допускают возможность настройки на тот или иной режим. В частности, интегрированная среда Turbo Pascal позволяет устанавливать как обычный, так и уплотненный режимы дисплея (переключатель Screen sizes  диалоговом окне Preferences – команда Options/Environment/Preferences).

При инициализации программы, использующей модуль CRT, значение текстового режима устанавливается автоматически, исходя из реальной аппаратуры компьютера.Текущее значение установленного режима всегда хранится в переменной LastMode из интерфейса модуля CRT. Даже в случае переключения в графический режим это значение сохраняется и может быть использовано для возврата в текстовый режим, например, TextMode (LastMode).

Рассмотрим подробнее процедуры и функции модуля CRT.

Процедуры управления цветом

TextBackGround(Color : byte) – процедура задает цвет фона Color.

TextColor(Color : byte) – процедура задает цвет символов Color.

HighVideo – процедура устанавливает высокую яркость символов (заменяет цвета 0-7 на цвета 8-15), выводимые далее на экран.

LowVideo – процедура устанавливает малую яркость символов (заменяет цвета 8-15 на цвета 0-7), выводимые далее на экран.

NormVideo - процедура устанавливает первоначальную яркость символов, выводимые далее на экран.

Управление цветом

Использование цвета при разработке видеоинтерфейсов программ, отображение результатов и т.п. является важным средством повышения удобства работы с программой, улучшения восприятия информации и в конечном итоге служит повышению производительности труда. Разумное и продуманное использование цвета может заметно повысить привлекательность программы и более наглядно выявить ее преимущества.

Цветные дисплеи персональных компьютеров позволяют в текстовом режиме выводить информацию на экран, используя фиксированный набор цветов. Различаются цвета самих символов и цвета фонов символов – областей знакомест, не занятых символами.Возможно независимое управление цветовыми характеристиками всех знакомест экрана. Базовый набор допустимых цветов составляет 16 цветов для символов, 8 цветов из этого набора можно задавать для фонов символов. Цвета кодируются последовательными числами от 0 до 15; эта кодировка определяется устройством дисплейной аппаратуры компьютера и поэтому не зависит от языка программирования.

Сonst

   {Коды, допустимые для текста и фона}

   Black

=0;
{Черный}

   Blue

=1;
{Синий

   Green

=2;
{Зеленый}

   Cyan

=3;
{Голубой}

   Red

=4;
{Красный}

   Maganta

=5;
{Пурпурный}

   Brown

=6;
{Коричневый}

   LightGray
=7;
{Серый}

   {Коды, допустимые только для текста}

   DarkGray
=8;
{Темно-серый}

   LightBlue
=9;
{Ярко-синий}

   LightGreen
=10;
{Ярко-зеленый}

   LightCyan
=11;
{Ярко-голубой}

   LightRed

=12;
{Ярко-красный}

   LightMaganta
=13;
{Ярко-пурпурный}

   Yellow

=14;
{Ярко-желтый}

   White

=15;
{Белый}

Для того, чтобы вывести на экран текст с определенными цветовыми характеристиками, в языке Turbo Pascal принята следующая техника: сначала устанавливаются текущие цветовые атрибуты, после чего все тексты, выводимые процедурами write и writeln на экран, используют заданные текущие атрибуты, до тех пор, пока они не будут переопределены. Текущий цвет и фон символов задаются процедурами TextColor и ТextBackground, которые устанавливают соответственно, текущий цвет для выводимых символов и текущий фон для них. Эти процедуры вызываются с одним параметром, который представляет значение нужного цвета (можно использовать одну из представленных выше констант, ее числовое значение или произвольное выражение, вырабатывающее целое значение в диапазоне от 0 до 15).

Помимо цвета, для символов можно задавать периодическое изменение яркости свечения, которое воспринимается как мерцание символа (мерцает только символ; фон остается неизменным). Как правило, кратковременная установка мерцания используется для вывода важных сообщений, имитации работающих электронных часов и т.п. Чтобы задать мерцание для выводимого далее текста, нужно увеличить значение параметра процедуры TextColor на 128; для большей наглядности этой операции наряду с кодами цветов в модуле CRT имеется соответствующая константа мерцания:

Const
   Blink =128;

Задание. Наберите на компьютере и внимательно рассмотрите представленную ниже простую демонстрационную программу, иллюстрирующую технику управления цветами.

Program DemoColors;

Uses
   Crt;

Begin

   writeln('По умолчанию выводятся белые символы на черном фоне');

   TextColor(Yellow);

   ТextBackground(Blue);

   writeln('Желтые символы на синем фоне');

   TextColor(Red);

   ТextBackground(LightGray);

   writeln('Красные символы на сером фоне');

   TextColor(White+Blink);

   ТextBackground(Red);

   writeln('Белый мерцающий текст на красном фоне');

   TextColor(Yellow);

   ТextBackground(Blue);

   writeln('A');

   TextColor(Red);

   ТextBackground(LightGray);

   writeln('B');

   TextColor(White);

   ТextBackground(Red);

   writeln('C');

End.

Процедуры и функции работы с экраном и курсором

ClrEol – процедура удаляет все символы от курсора (включительно) до конца строки, заполняя этот участок строки цветом фона.

ClrScr – процедура очищает текущее окно, заполняя его цветом фона и помещает курсор в его верхний левый угол с координатами (1, 1).

DelLine – процедура удаляет строку, в которой находится курсор.

GoToXY(X, Y : byte) – процедура перемещает курсор к элементу экрана с заданными координатами X и Y. Если хотя бы одна координата недопустима, процедура не выполняется.

InsLine – процедура вставляет пустую строку на экране в месте расположения курсора и заполняет ее цветом фона.

WhereX (Function WhereX : byte) – возвращает значение горизонтальной кооординаты.

WhereY(Function WhereY : byte) – возвращает значение вертикальной кооординаты.

Управление курсором

В текстовом режиме на экране, как правило, присутствует курсор – мерцающий служебный элемент, назначением которого является указание позиции на экране, начиная с которой будет производиться вывод или ввод информации. Положение курсора на экране можно проверять и изменять, используя подпрограммы CRT. Чтобы установить курсор в некоторой конкретной позиции, используется процедура GoToXY, два параметра которой определяют позицию знакоместа на экране, в которую будет перемещен курсор. Первый параметр задает номер позиции по горизонтали (номер колонки), второй параметр – номер по вертикали (номер строки). Строки и колонки нумеруются от единицы.

Например, вызов процедуры GoToXY(1,1) переместит курсор в крайнюю левую верхнюю позицию экрана, а вызов процедуры GoToXY(80,25) – в нижнюю правую.

Заметим, что второй вызов даст такой эффект только в том случае, когда текущий текстовый режим равен BW80, CO80, Mono. В общем случае необходимо учитывать симло строк и столбцов для конкретного режима. Если, например, действует режим CO80+Font8x8 для адаптера VGA (50 строк по 80 символов), то приведенный выше вызов процедуры переместит курсор примерно в середину правой крайней колонки.

Чтобы определить текущее положение курсора, можно использовать две функции без параметров - WhereX и WhereY, которые как видно из их названий, возвращают, соответственно, горизонтальную и вертикальную координаты курсора.

Задание. Наберите, протестируйте и рассмотрите небольшую иллюстративную программу, приведенную ниже.

Program Count;

Uses

   Crt;

Var

   i : word;

Begin

   GotoXY(1,WhereY);

   write('                                 ');

   for i := 1 to 1000 do

      begin

          GotoXY(1,WhereY);

          write(i);

          Delay(100);

      end;

End.

Прием, используемый в этой программе, можно применять для нужд трассировки. Например, при обработке очень больших файлов, которая может продолжаться довольно долгое время, имеет смысл выводить "успокаивающую" информацию, наглядно показывающую пользователю, что компьютер не "завис" (некоторые системы выдают процент сделанной работы или что-либо подобное).

Текстовый режим. Модуль CRT. Работа с окнами. Управление звуком

Модуль CRT позволяет работать не только со всем экраном, но и выделять в нем прямоугольные окна. Любое окно задаётся своим верхним левым и правым нижним углами. Эти углы, так же как и положение любого объекта на экране, задаются двумя координатами: Х и Y. В качестве координаты Х выступает номер позиции в строке (нумерация начинается с 1 и идет слева направо), а в качестве координаты Y - номер строки (нумерация начинается с 1 и идет сверху вниз). При работе в окне координаты отсчитываются от левого верхнего угла окна. При запуске программы выделенное окно совпадает по размеру со всем экраном. Если режим работы адаптера 25 строк по 80 позиций, то координаты такого окна (1, 1) и (80, 25).

Window(X1, Y1, X2, Y2 : byte) – определяет текстовое окно на экране.

Смысл выполнения процедуры заключается в том, что все последующие действия по выводу информации действуют в пределах данного окна. Кроме того, системные процедуры, работающие с курсором - GotoXY, WhereX, WhereY, также интерпретируют свои параметры относительно левого верхнего угла последнего установленного окна. Далее, процедура ClrScr, заполняет пробелами не весь экран, а текущее окно.

Значения параметров процедуры Window выбираются очевидным образом; например, вызов

Window(35,11,45,13);

определяет окно высотой в две строки и длиной в десять символов примерно посередине экрана.

После выполнения процедуры Window курсор устанавливается в позицию окна с координатами (1,1). Кроме этого, эффект данной процедуры на экране никак не проявляется. Результат ее работы бывает виден только при выполнении каких-либо процедур вывода или управления курсором. Хорошей иллюстрацией механизма окон является следующая простая программа, которая в цикле производит создание очередного окна, случайным образом задавая его размеры м местоположение, а затем выводит в него серию случайных символов. 

Задание. Наберите и протестируйте предложенную программу. Дополните ее своими операторами и комментарием. В целях большей наглядности употребите процедуру Delay и увеличьте длительность цикла вывода символов.

Program Windows;

Uses

   Crt;

Procedure RandomWindow;

Var

   X1, Y1, X2, Y2 : byte;

Begin

   X1 := Random(40)+1;

   Y1 := Random(12)+1;

   X2 := X1+Random(40);

   Y2 := Y2+Random(12);

   TextColor(Random(15));

   Window(X1, Y1, X2, Y2);

   ClrScr;

End;

Procedure WorkWindow;

Var

   i : integer;

Begin

   for i := 0 to Random(1000)+500 do

       write(Chr(Random(224)+32));

End;

Procedure Finish;

Var

   i : integer;

Begin

   Window(1, 1, 80, 25);

   ClrScr;

   if ReadKey = #0

      then

         ReadKey;

End;

Begin

   ClrScr;

   Randomize;

   repeat

       RandomWindow;

       WorkWindow

   until KeyPressed;

   Finish;

End.

Модуль CRT содержит две интерфейсные переменные типа word, которые хранят в упакованном виде координаты двух углов текущего активного окна:

Var

   WindMin : word; {координаты левого верхнего угла окна}

   WindMax : word; {координаты правого нижнего угла окна}

Посредством стандартных функций Hi и Lo можно извлечь из этих переменных горизонтальные и вертикальные координаты углов. Наличие этих переменных избавляет от необходимости хранить информацию об активном окне в самой программе.

Наблюдая работу программы Windows, можно заметить, что, когда в процессе вывода производится заполнение последней позиции окна, его содержимое сдвигается на одну строку вверх (при этом верхняя строка пропадает) и, тем самым, нижняя строка освобождается для последующего вывода. Этот механизм является естественным обобщением знакомых правил работы процедур write и writeln на случай вывода в окна. В модуле Crt имеются две дополнительные процедуры, позволяющие несколько расширить и усилить возможности прокручивания("скроллинга") содержимого окон –  DelLine и InsLine.

Процедура DelLine удаляет строку окна, на которой находится курсор, и перемещает все строки ниже этой строки на одну строку вверх. При этом нижняя строка очищается и тем самым допускает ввод очередной порции информации. На основе этой процедуры удобно задавать прокрутку текстов в окне в направлении снизу вверх. 

Задание. Наберите и протестируйте предложенную программу. Дополните ее своими операторами и комментарием. 

Program ScrollUp;

Uses

   Crt;

Var

   i : word;

Begin

   Window(5,10,65,14);

   ClrScr;

   writeln('Cтрока 1');

   writeln('Cтрока 2');

   writeln('Cтрока 3');

   writeln('Cтрока 4');

   writeln('Cтрока 5');

   for i := 6 to 26 do

      begin

          GotoXY(1,1);

          DelLine;

          GotoXY(1,5);

          write('Строка ',i);

          Delay(300);

      end;

   Delay(500);

   Window(1, 1, 80, 25);

   ClrScr;

End.

Процедура InsLine вставляет пустую строку в позиции курсора. При этом все строки окна, расположенные ниже новой строки, сдвигаются вниз (нижняя строка пропадает).Легко видеть, что эта процедура позволяет организовать прокрутку содержимого окна (или части окна) в направлении сверху вниз. 

Задание. Наберите и протестируйте предложенную программу. Дополните ее своими операторами и комментарием. 

Program ScrollDown;

Uses

   Crt;

Var

   i : word;

Begin

   Window(5,10,65,19);

   ClrScr;

   writeln('Данные строки окна');

   writeln('остаются неизменными.');

   writeln('В пяти нижних строках');

   writeln('организована прокрутка');

   writeln('в направлении сверху вниз');

   for i := 1 to 40 do

      begin

          GotoXY(1,6);

          InsLine;

          write('Строка ',i);

          Delay(300);

      end;

   Delay(500);

   Window(1, 1, 80, 25);

   ClrScr;

End.

Процедуры управления звуком

Sound(Hz : word) – процедура включает источник звука с частотой Hz герц.

NoSound – процедура выключает источник звука.

Процедуры для работы со звуком позволяют получить доступ ко встроенному динамику компьютера. Процедура Sound включает динамик, который начинает генерировать звук с заданной в герцах частотой тона. Это никак не отражается на работе программы, которая не будет остановлена и может продолжать нормально работать одновременно с подачей звуковых сигналов до тех пор, пока не встретится вызов процедуры NoSound, которая отключает динамик.

Процедуры работы со звуком надо использовать аккуратно. Если после процедуры Sound не вызвать процедуру NoSound, то звук будет производится после завершения программы, и, чтобы его отключить, придется либо перезагружать компьютер, либо заново перекомпилировать и запустить программу, предварительно добавив в нее вызов процедуры NoSound.

Процедура NoSound может следовать после нескольких процедур Sound, которые изменяют частоту тона.

Очень часто процедуры Sound и NoSound используются совместно с процедурой Delay(ms : word), которая останавливает выполнение программы на указанное в параметре время. 

Процедура Delay использует для отсчета времени встроенную микросхему таймера, которая вычисляет десятые и сотые доли секунды недостаточно точно, поэтому не следует применять эту процедуру для точного вычисления времени.

Рассмотрите пример использования звуковых процедур. Программа генерирует последовательность звуков, представляющих обычную гамму.

Program Gamma;

Uses

   Crt;

Const

   Nota : Array [0..7] of word = (262, 294, 330, 349, 392, 440, 494, 524);

Var

   i : byte;

Begin

   GotoXY(35,5);

   write('Гамма');

   repeat

       for i := 0 to 7 do

           begin

               Sound(Nota[i]);

               Delay(500);

               NoSound;

           end;

   until KeyPressed;

End.

Функции работы с клавиатурой
KeyPressed (Function KeyPressed : boolean) – определяет, была ли нажата клавиша на клавиатуре.

ReadKey (Function ReadKey : char) – читает значение нажатой клавиши.

Эти функции были подробно рассмотрены в теме "Циклы" первого года обучения. Поэтому здесь мы останавливаться на их изучении не будем.

Внимание! Дополните любую решенную Вами задачу рассмотренными в этом занятии процедурами и функциями.

